Solar panel battery charger

CONTENTS

1. What is a solar panel? ……………………………………………1
2. Modeling a solar cell in LTSpice………………………………3
   2.1. Why Use a Voltage Source for simulating the illumination?...4
   2.2. Simulating the solar cell…………………………………….5
   2.3. Making a symbol for solar cell…………………………...8
   2.4. Making a solar cell using parameters from a datasheet…10
   2.5. Making a schematic using just symbols………………...15
3. The charging circuit……………………………………………...19
   3.1. Modeling the charging of a battery……………………..22
4. Bibliography…………………………………………………..24

1. What is a solar panel?

A solar panel is built with multiple individual solar cells that can be put in series or in parallel, or both, depending on what current or what voltage we want to obtain. The use of a solar panel is that it converts light energy into electrical energy.

To understand how a solar panel functions is very important to know that a solar panel is formed with solar cells.

When we think about a solar cell, we can imagine a diode that has extra dimensions. So, a solar cell has one or more junctions and when we expose those junctions to light a certain voltage will be produced. To manufacture a solar cell, silicon is very often used. We use silicon as an intrinsic region situated between p dopped material and a n dopped material. When the solar radiation hits the silicon a bunch of holes and electrons will appear, it will create an electric field inside the semiconductor, the holes and the electrons will go in different paths and will form an electric current. So, we can approximate this configuration with a circuit that has a source of current (photovoltaic current, basically the diode current), a diode and some load resistors.

That being said, the typical solar cell model it looks like this:



Fig. 1. Solar cell circuit

Where R_S represents the various series elements interconnections between multiple junctions or wiring, R_{SH} (shunt resistance) for various internal losses or leakages.

The parameters needed to define this model are the values of the resistors (R_S and R_{SH}) and the current generated by the current source (I_{PH} – the current conversion from light energy, is depended to the light intensity to witch the panel is exposed to, or temperature). The diode is defined by the parameters used in Shockley equation:

$$I_d = I_o\left(e^{\frac{qV_d}{nakT}} - 1\right) \quad (1)$$

I_o → dark saturation current.

a → ideality factor (how close is the response of this diode to an ideal diode).

n → the number of junctions in series in your modeled cell.

2. Modeling a solar cell in LTSpice

We will use the solar cell model discussed in the previous pages for implementing and testing in LTSpice.

The schematic should look like this:



Fig. 2. LTSpice schematic for a solar cell

Illu - the illumination net that will be used to set the illumination exposure to the solar panel.

B1 - a behavioral source is used to represent the current generated by the photovoltaic effect, which is dependent on illumination. To define this behavioral source, we use an equation that is a simplified way to simulate the illumination-dependent current generated by a solar cell. Here's a breakdown of the equation:

→ V(illu): This represents the illumination or irradiance level that the solar cell receives, typically in units of watts per square meter (W/m^2). The "illu" net is a voltage node that contains information about the intensity of the illumination.

→ 1000: This is the normalization factor for irradiance. In most solar cell simulations, the standard reference irradiance is 1000 W/m^2, which corresponds to "full sun" under standard test conditions (STC). The division by 1000 normalizes the illumination so that at 1000 W/m^2, the photovoltaic current is simply I_{PH}, without scaling.

→ I_{PH}: This is the photocurrent generated by the solar cell when it is exposed to full standard illumination (1000 W/m^2). The value of I_{PH} depends on the solar cell's characteristics, such as the area of the cell and its efficiency.

D1 – solar diode, for this we have taken the built-in model in LTSpice, and this will be using three parameters. We have the I_o parameter which is attributed to the built-in saturation current, the ideality factor (a) and number of cells (n) - a_n, will be used as a single parameter attributed to n (from the built-in model) - the emission coefficient, tref represents the temperature at which these parameters have been calculated, in our case it will be equal to the global temperature of the simulation.

2.1. Why Use a Voltage Source for simulating the illumination?

The voltage source at the node labeled illu does not represent an electrical voltage in the traditional sense but is being used to numerically represent the irradiance in watts per square meter (W/m^2).

LTSpice is built to handle electrical quantities, so to simulate non-electrical phenomena like illumination, we map those to electrical analogs. In this case, voltage is used as a placeholder for irradiance.

This means that:

The voltage at the node illu represents the irradiance in W/m^2. For example, setting the voltage source to 500 V would correspond to an irradiance of 500 W/m^2.

2.2. Simulating the solar cell

We made a transient analysis of this circuit, and the results was the following:



Fig. 2.2. Simulation results

Green curve:

This is the I-V characteristic of the solar cell. It shows how the current through the solar cell varies with the voltage across it.

Short-circuit current (I_{SC}): is the maximum current that the solar cell can deliver when the terminals are shorted (i.e., when V=0V). The equation for I_{SC}:

$$\mathbf{I_{sc} = I_{ph} - I_o\left(e^{\frac{qV_{oc}}{nV_T}} - 1\right)} \quad (2)$$

At zero voltage, the current reaches a maximum value (~1A in simulation), which represents the short-circuit current. This is the maximum current when the solar cell is not loaded (V = 0). From the solar cell current equation: $I_{SC}=I_{PH}$. In our LTSpice simulation, we have set:

.param iph = 1

Open-circuit voltage (V_{OC}): is the maximum voltage across the solar cell when no current is drawn from it (i.e., I=0). The equation for V_{OC} is:

$$\mathbf{V_{oc} = nV_T \ln\left(\frac{I_{ph}}{Io} + 1\right)} \quad (3)$$

Replacing the notations with the numeric value V_{OC} = 3.59V.

In the simulation the point where the current drops to zero (around 3.5V in our case) represents the open-circuit voltage. This is the maximum voltage when the solar cell is not delivering any current (I = 0).

The curve between these two extremes represents the solar cell's normal operation. The current decreases as the voltage increases, due to the diode effect and the influence of series resistance.

The link from the equations Solar Cell Equation - an overview | ScienceDirect Topics

Blue curve (Power vs. Voltage):

This curve represents the output power of the solar cell as a function of voltage, calculated as P=I×V.

The power increases as the voltage rises, reaches a peak (around 1.8W in our case), and then drops as the voltage approaches the open-circuit voltage.

This peak corresponds to the **maximum power point (MPP)**, where the solar cell delivers maximum power to the load. This point is critical for real-world solar cell applications because you want to operate the solar cell as close to the MPP as possible to maximize energy harvesting.

If we ask ourselves why I2 and V+ (we put V+ on the X axes) are the values that are interesting us those are the reasons:

→**Behavioral Source (B1)**: This source represents only the *photocurrent* part, which is the current generated based on illumination. However, the total output of a solar cell is influenced by other factors such as series resistance (rs_t), shunt resistance (rsh_t), diode characteristics, and load conditions.

→**Current Source (I1)**: Testing the current here (which likely represents the current flowing through the load) gives you the true output current of the entire system. It considers not just the illumination-dependent current, but also the effects of:

The diode (D1), which models the recombination losses and junction behavior of the solar cell.

The series resistance (R=rs_t), which models losses in the cell material.

The shunt resistance (R1=rsh_t), which accounts for leakage currents.

The load itself, which represents the actual working condition of the solar panel.

- →**V+** is the output voltage, so it expresses the value of the voltage after the effects of the other components from schematic.

2.3. Making a symbol for solar cell

Next step of our process is to take that from the simulator and put it into a library file by making a symbol for our solar cell.

To do this we will open the spice netlist file and copy the information into a txt file. This is what the txt file should contain for this moment:

```
.subckt PVbasic v+ v- illu
B1 v- N001 I=v(illu)/1000*Iph
D1 N001 v- SolarDiode
R1 N001 v- {Rsh}
R2 v+ N001 {Rs}
R3 illu 0 100Meg
.model solardiode d(is=io n=a_n tref=temp)
.ends
```

So, we will define a subcircuit that is called PVbasic, and it has three interface pins v+, v- and illu. Now to be actually able to use this component we need to define a new component where all of the various parameters that are needed inside of the model are defined, and we will write that also in the same txt file at the beginning. The txt file will look like this:

```
*
.subckt testcell1 1 2 3
xul 1 2 3 PVbasic rs=1 rsh=10k iph=1 io=1e-6 a_n=10
.ends
*

.subckt PVbasic v+ v- illu
B1 v- N001 I=v(illu)/1000*Iph
```

```
D1 N001 v- SolarDiode
R1 N001 v- {Rsh}
R2 v+ N001 {Rs}
R3 illu 0 100Meg
.model solardiode d(is=io n=a_n tref=temp)
.ends
```

Here we defined the subcircuit testcell1 that has three pins connected to nets 1, 2 and 3 and under this component we have another component xu1 with the same pins connected to the same nets which is a type component PVbasic that has the five parameters attributed to it.

The next step is to open the txt file in LTSpice, click right on the .subckt testcell1 1 2 3 and choose Create a symbol option.



Fig. 2.3.1. Solar cell symbol

From the library file we know that pin 1 is v+, pin 2 is v- and pin 3 is illu, after making these changes we can save the symbol in our library and use it into a schematic.



Fig. 2.3.2. Circuit with Solar cell symbol



Fig. 2.3.3. The simulation results

After this simulation we can see that the results are the same as in the previous case so the symbol works properly.

2.4. Making a solar cell using parameters from a datasheet

In the previous case we used some parameters there are not that easy to obtain in real life. Now we want to use just what a datasheet from a solar panel has to offer.

In order to do that we need to start with a specific algorithm to work with. From different science paper we get these formulas:

$$\mathbf{V_{oc_T} = V_{oc}\left(1 - T_{k_{V_{oc}}}(25 - TEMP)\right)} \qquad (4)$$

TEMP → the temperature use at the simulation

T_{k_Voc} → temperature coefficient at open circuit voltage

V_{oc} → open circuit voltage

$$\mathbf{I_{sc_T} = I_{sc}\left(1 - T_{k_{I_{sc}}}(25 - TEMP)\right)} \qquad (5)$$

T_{k_Isc} → temperature coefficient at short circuit current

I_{sc} → short circuit current

$$\mathbf{V_{MP_T} = V_{MP}\left(1 - T_{k_{V_{MP}}}(25 - TEMP)\right)} \qquad (6)$$

T_{k_VMP} → temperature coefficient at maximum power voltage

V_{MP} →maximum power voltage

$$\mathbf{I_{MP_T} = I_{MP}\left(1 - T_{k_{I_{MP}}}(25 - TEMP)\right)} \qquad (7)$$

T_{k_IMP} → temperature coefficient at maximum power current

I_{MP} →maximum power current

These values will help us to calculate the parameters for our simulation model.

The parameters for our simulation model are:

I_{o_T} → dark saturation current

a → ideality factor

n → number of cells

R_s → series resistance

R_{sh} → shunt resistance

I_{PH} → photo current

We will choose our ideality factor to be 1.3, usually is used a value between 1 and 2. Assuming that we don’t know how many cell we have n will have this value $V_{oc}/0.7$ (the typical forward voltage for a silicon diode), so we can make a single coefficient named a_n = 1.3 * $V_{oc}/0.7$.

$$\mathbf{R_{S_T}} = \frac{\mathbf{V_{OC_T}} - \mathbf{V_{MP_T}}}{\mathbf{16 I_{MP_T}}} \qquad (8) \qquad\qquad \mathbf{R_{Sh_T}} = \frac{\mathbf{5 V_{MP_T}}}{\mathbf{I_{SC_T}} - \mathbf{I_{MP_T}}} \qquad (9)$$

$$\mathbf{I_{PH_T}} = \mathbf{I_{SC_T}} \frac{\mathbf{R_{SH_T}} + \mathbf{R_{S_T}}}{\mathbf{R_{SH_T}}} \quad (10) \qquad\qquad \mathbf{V_{T_T}} = \frac{\mathbf{k\ TEMP(kelvin)}}{\mathbf{q}} \quad (11)$$

$$\mathbf{I_{O_T}} = \frac{(\mathbf{R_{S_T}} + \mathbf{R_{Sh_T}})\mathbf{I_{SC_T}} - \mathbf{V_{OC_T}}}{\mathbf{R_{SH_T}} \mathbf{e}^{\frac{\mathbf{V_{oc_T}}}{\mathbf{a_n V_{T_T}}}}} \qquad (12)$$

The science papers used: oa.upm.es/30687/1/2014enef.pdf

161058299.pdf (core.ac.uk)

65AB83859477 (academicjournals.org)

Now using the datasheet from a solar panel SEP300-320.pdf we will model it in LTSpice with the parameters from this document. Also, the datasheet has graphs with which we can compare our results.



Fig 2.4.1. Solar cell model

By simulation the current I-V characteristic is:



Fig 2.4.2. The I-V characteristic resulted by simulation

We obtain these results by stepping the illumination with 200 600 and 1000W. Compared to the characteristic from datasheet (the link from that was pasted previously) the results are approximative the same.



Fig 2.4.3. The I-V characteristic from datasheet



Fig. 2.4.4. The simulation for I-V Curves at different temperatures



Fig. 2.4.5. The simulation for I-V Curves at different temperatures from datasheet

2.5. Making a schematic using just symbols

Like in the previous case the next step is to take the Spice Netlist from this schematic and put it into a text file.

The text file should look like this:

```
*
.subckt SEP300W 1 2 3 ; http://sunceco.com/wp-content/uploads/2017/01/SEP300-320.pdf
xu1 1 2 3 PV_cplx voc=44.71 isc=8.947 vmp=37.23 imp=8.06 tk_voc=-0.0034 tk_isc=0.0005
.ends
.subckt CL_SM10P 1 2 3; https://www.tme.eu/Document/f73597b9cc5801bdd87f2781fa4ee352/CL-SM10P.pdf
xu 1 2 3 PV_cplx voc=22.6 isc=0.59 vmp=18.2 imp=0.55 tk_voc=-0.004 tk_isc=0.00065
.ends
.subckt MP3_25 1 2 3 ; https://ro.mouser.com/datasheet/2/1009/Electronic_Component_Spec_Sheet_Cla_77DEA84523C82-1658524.pdf
xu 1 2 3 PV_cplx voc=4.1 isc=0.035 vmp=3 imp=0.007 tk_voc=-0.004 tk_isc=0.00065
.ends
```

```
*
*PhotoVoltaic Cell model
*
* necessary parameters
*voc - open circuit voltage
*isc - short circuit current
*vmp - maximum power voltage
*imp maximum power current
*tk_voc - open circuit voltage temperature coeffiecient
*tk_isc - short circuit current temperature coefficient
*
.subckt PV_cplx v+ v- illu
D1 N002 v- pvdiode2
Rsh N002 v- R=rsh_t
Rs N002 v+ R=rs_t
Bph v- N002 I=ipv_t*v(illu)/1000
R5 illu 0 100meg
.param voc_t voc*(1-tk_voc*(25-temp))
.param isc_t isc*(1-tk_isc*(25-temp))
.param vmp_t vmp*(1-tk_voc*(25-temp))
.param imp_t imp*(1-tk_isc*(25-temp))
.param rs_t (voc_t-vmp_t)/(16*imp_t) ; 2 instead of 16 in original source
.param rsh_t 5*vmp_t/(isc_t-imp_t) ; 1 instead of 5 in original source
.model PVDiode2 D(Is=io_t N=a_n Tnom=temp)
.param io_t ((rs_t+rsh_t)*isc_t-voc_t)/(rsh_t*exp(voc_t/(a_n*vt_t)))
.param vt_t (1.38e-23*(273+temp))/1.6e-19
.param ipv_t isc_t*(rsh_t+rs_t)/rsh_t
.param a_n 1.3*voc/0.7
*
.ends
```

We have our model in the bottom of the txt file, it is named PV_cplx(photovoltaic complex model), it has the same three inputs as in the previous case, it has 5 components and then the various calculations needed for the parameters.

At the top of the txt file, we have three commercially available solar panels with various values that define their behavior.

Now we can copy the txt file in our schematic as a spice directive, we will use for our test circuits the symbol that we previously implemented, and we can test it.



Fig. 2.5.1. Schematic for 3 different solar panels

By testing it the results can be compared to the datasheet file for each solar panel.

At this moment we can put together two solar cells, and we can testing it.



Fig. 2.5.2. Schematic for two solar cells in series



Fig. 2.5.3. Simulation results

From this simulation we can see that the current has the same value, but the voltage has been doubled.

3. The charging circuit

The next step in our project is to make a solar battery charger circuit.

The schematic should look like this:



Fig. 3.1. The schematic for the solar panel battery charger

We are trying to charge a battery of 12V and 1.3Ah from a solar panel. To do that, we use two diodes in circuit, D1 and D2, a voltage regulator LT317, a capacitor, two resistors and a potentiometer.

D1 is used to prevent reverse polarity – "Placing a reverse polarity protection diode in series with the electrical supply line provides a "shut-off" mechanism to halt the voltage flow. It doesn't fix the reverse polarity, but it does stop it from doing more harm.", more information [PCB Design Guide for Reverse Polarity Protection (matric.com)](matric.com).

D2 is a Schottky diode used to protect the LM317 and panel from reverse voltage generated by the battery when it is not charging. This diode has a forward current of 3 A and a V_{RRM} of 50 V (VRRM -

maximum repetitive reverse voltage, the maximum amount of voltage the diode can withstand in reverse-bias mode, in repeated pulses).

C1 protects from static discharge (the release of static electricity when two objects come into contact).

LT317 is a voltage regulator capable of supplying more than 1.5 A over an output-voltage range of 1.25 V to 37 V. It requires only two external resistors to set the output voltage. (A voltage regulator is usually used to maintain the voltage of a power source within acceptable limits).

We will use the first solar panel that we modeled and a lead-acid battery with 12 V and 1.3 Ah. The voltage specified in the battery cycle use should be between 14.4-15 V and the maximum charge current has a value of 0.39 A.

The datasheet battery: Oracle spec-HD1213 (1).pdf (oraclebattery.com) .

As we said, the battery needs an input voltage between 14.4-15V so the output voltage of the regulator should have a value of 15 V by adding the forward voltage from the Schottky diode. Also, we need to pay attention to the current that passes through the diode D2. We need to consider a minimum current of 130 mA (10% from 1.3 Ah) and a maximum one of 390 mA.

To obtain those values we used LM317 Calculator (Voltage Source) - Daumemo for calculating the resistors. By the calculations that the program did for us R1 should have a value of 220 **Ω**, and R2 with the potentiometer should have together 2.36 k**Ω** with an input voltage for regulator between 17.7 V to 40 V. We made some approximations, and this is our results by the simulation:



Fig. 3.2. The simulation results for voltage

The output voltage of the solar panel (that is the input voltage for the voltage regulator + the forward voltage of the diode D1) is not the maximum power voltage because the load has moved the operating point to the I-V characteristic.



Fig. 3.3. The simulation results for current

Considering that the value of the current that passes through the battery has a value of 164.5mA our battery should be fully charged in 7.9h (1.3Ah/164.5mA).

3.1. Modeling the charging of a battery

To simulate how the battery is charging we will approximate the circuit of the battery with a DC voltage source in series with a small resistance and a capacitor.

For this we will assume that the minimum voltage for the battery (when the battery is fully discharged) is 10.5 V (we chose this value based on industry standards and best practices for battery care. This minimum voltage prevents over-discharge, which can harm the battery.). Also, the maximum voltage will be 13.8 V – the voltage when the battery is fully charged (in datasheet is a value between 13.8-14.4 V). The intern resistance of this battery is 95 mΩ. To calculate the value of the capacitor we will use this formula:

$$\boldsymbol{C = \frac{Ah \times 3600}{V}} \qquad \mathbf{(13)}$$

Where:

- C is the capacitance in farads,
- Ah is the amp-hour rating of the battery,
- V is voltage difference between the maximum and the minimum battery voltages,
- 3600 is the conversion factor from hours to seconds.

So, after replacing the values the capacitor will have a value of 1418.18 F.

The schematic will look like this:



Fig. 3.1.1. The model of a battery

Practically we will pass the battery through a process from fully discharged to fully charged using the initial condition of 10.5 V and a DC voltage source of 13.8 V.

The simulation result is:



Fig. 3.1.2. The charging process

4. Bibliography

[1] Solar Cell Circuit Model Explained (youtube.com)

[2] Solar Cell Circuit (with Load attached) (youtube.com)

[3] Modeling Photovoltaic Cells - Theory 1/2 (youtube.com)

[4] Modeling Photovoltaic Cells - LTspice model part 2/2 (youtube.com)

[5] Solar Battery Charger Circuit using LM317 Voltage Regulator (electronicshub.org)